Taylor Expansions about t=0

In[70]:=
```
yhTaylor[n_] =
  y[0] + Sum[ (D[f[t], {t, k - 1}] /. t → 0) h^k/k! , {k, n}] (* y(h) in terms of f(t) *)
```

Out[70]= $\sum_{k}^{n} \frac{h^k f^{(-1+k)}[0]}{k!} + y[0]$

In[71]:=
```
yhTaylor[5]
```

Out[71]= $h\, f[0] + y[0] + \frac{1}{2} h^2 f'[0] + \frac{1}{6} h^3 f''[0] + \frac{1}{24} h^4 f^{(3)}[0] + \frac{1}{120} h^5 f^{(4)}[0]$

In[72]:=
```
ftTaylor[t_, n_] :=
  Sum[ (D[f[τ], {τ, k}] /. τ → 0) t^k/k! , {k, 0, n}] (* f(t) expansion *)
```

In[73]:=
```
ftTaylor[h, 5]
```

Out[73]= $f[0] + h\, f'[0] + \frac{1}{2} h^2 f''[0] + \frac{1}{6} h^3 f^{(3)}[0] + \frac{1}{24} h^4 f^{(4)}[0] + \frac{1}{120} h^5 f^{(5)}[0]$

Forward Euler O(h^2)

In[74]:=
```
yhFwdEuler = y[0] + h f[0]
```

Out[74]= $h\, f[0] + y[0]$

In[75]:=
```
yhFwdEuler - yhTaylor[2]
```

Out[75]= $-\frac{1}{2} h^2 f'[0]$

Backward Euler O(h^2)

In[76]:=
```
yhBwdEuler = Expand[y[0] + h ftTaylor[h, 1]]
```

Out[76]= $h\, f[0] + y[0] + h^2 f'[0]$

In[77]:=
```
yhBwdEuler - yhTaylor[2]
```

Out[77]= $\frac{1}{2} h^2 f'[0]$

Trapezoidal Method O(h^3)

In[78]:=
```
ftTaylor[h, 2]
```

Out[78]= $f[0] + h f'[0] + \frac{1}{2} h^2 f''[0]$

In[79]:=
```
yhTrapezoid = Expand[y[0] + h/2 (f[0] + ftTaylor[h, 2])]
```

Out[79]= $h f[0] + y[0] + \frac{1}{2} h^2 f'[0] + \frac{1}{4} h^3 f''[0]$

In[80]:=
```
Series[ yhTrapezoid - yhTaylor[5], {h, 0, 3}]
```

Out[80]= $\frac{1}{12} f''[0] h^3 + O[h]^4$

Midpoint Method O(h^3)

Runge-Kutta Methods

Use degree-n Taylor series expansion of F[y] about y_0:

In[83]:=
```
Fy0Taylor[Y_, n_] := F[y0] + Sum[ (D[F[X], {X, k}] /. X → y0) (Y - y0)^k / k! , {k, n}]
```

In[84]:=
```
Fy0Taylor[y0, 4]
```

Out[84]= $F[y_0]$

In[85]:=
```
Fy0Taylor[y0 + h v, 4]
```

Out[85]= $F[y_0] + h v F'[y_0] + \frac{1}{2} h^2 v^2 F''[y_0] + \frac{1}{6} h^3 v^3 F^{(3)}[y_0] + \frac{1}{24} h^4 v^4 F^{(4)}[y_0]$

Expanding $y^{(n)}(t)$ derivatives in terms of F and its gradients

In[86]:=
```
dyF[1] = F[v] (* y^(1) *)
```

Out[86]= $F[v]$

In[87]:=
```
dyF[2] = D[dyF[1], v] F[v] (* y^(2) *)
```

Out[87]= $F[v] F'[v]$

In[88]:=
```
dyF[3] = D[dyF[2], v] F[v] (* y^(3) *)
```

Out[88]= $F[v] \left(F'[v]^2 + F[v] F''[v]\right)$

In[89]:=
```
dyF[4] = D[dyF[3], v] F[v] (* y^(4) *)
```

Out[89]= $F[v] \left(F'[v] \left(F'[v]^2 + F[v] F''[v]\right) + F[v] \left(3 F'[v] F''[v] + F[v] F^{(3)}[v]\right)\right)$

In[90]:=
```
dyF[5] = D[dyF[4], v] F[v]  (* y^(5) *)
```

Out[90]=
$$F[v]\left(F'[v]\left(F'[v]\left(F'[v]^2+F[v]\,F''[v]\right)+F[v]\left(3\,F'[v]\,F''[v]+F[v]\,F^{(3)}[v]\right)\right)+F[v]\left(F''[v]\left(F'[v]^2+F[v]\,F''[v]\right)+2\,F'[v]\left(3\,F'[v]\,F''[v]+F[v]\,F^{(3)}[v]\right)+F[v]\left(3\,F''[v]^2+4\,F'[v]\,F^{(3)}[v]+F[v]\,F^{(4)}[v]\right)\right)\right)$$

In[91]:=
```
dyF[6] = D[dyF[5], v] F[v]  (* y^(6) *)
```

Out[91]=
$$F[v]\Big(F'[v]\Big(F'[v]\left(F'[v]\left(F'[v]^2+F[v]\,F''[v]\right)+F[v]\left(3\,F'[v]\,F''[v]+F[v]\,F^{(3)}[v]\right)\right)+F[v]\left(F''[v]\left(F'[v]^2+F[v]\,F''[v]\right)+2\,F'[v]\left(3\,F'[v]\,F''[v]+F[v]\,F^{(3)}[v]\right)+F[v]\left(3\,F''[v]^2+4\,F'[v]\,F^{(3)}[v]+F[v]\,F^{(4)}[v]\right)\right)\Big)+F[v]\Big(F''[v]\left(F'[v]\left(F'[v]^2+F[v]\,F''[v]\right)+F[v]\left(3\,F'[v]\,F''[v]+F[v]\,F^{(3)}[v]\right)\right)+2\,F'[v]\left(F''[v]\left(F'[v]^2+F[v]\,F''[v]\right)+2\,F'[v]\left(3\,F'[v]\,F''[v]+F[v]\,F^{(3)}[v]\right)+F[v]\left(3\,F''[v]^2+4\,F'[v]\,F^{(3)}[v]+F[v]\,F^{(4)}[v]\right)\right)+F[v]\left(\left(F'[v]^2+F[v]\,F''[v]\right)F^{(3)}[v]+3\,F''[v]\left(3\,F'[v]\,F''[v]+F[v]\,F^{(3)}[v]\right)+3\,F'[v]\left(3\,F''[v]^2+4\,F'[v]\,F^{(3)}[v]+F[v]\,F^{(4)}[v]\right)+F[v]\left(10\,F''[v]\,F^{(3)}[v]+5\,F'[v]\,F^{(4)}[v]+F[v]\,F^{(5)}[v]\right)\right)\Big)\Big)$$

Taylor series of y(h) about zero using F & its gradients for $y^{(n)}(t=0)$ derivatives:

In[92]:=
```
yhFExact[n_] := (y0 + Sum[dyF[i] h^i/i!, {i, 1, n}])
```

In[93]:=
```
yhF[n_] :=
 yhFExact[n] /. v → y0 (* Taylor series expansion of y(h) about t=0, y=y0 *)
```

In[94]:=
```
yhF[3] (* test *)
```

Out[94]=
$$h\,F[y_0]+y_0+\frac{1}{2}h^2\,F[y_0]\,F'[y_0]+\frac{1}{6}h^3\,F[y_0]\left(F'[y_0]^2+F[y_0]\,F''[y_0]\right)$$

Note: For each Runge-Kutta integrator, we will now compare its Taylor expansion (using F & gradients) to yhF[n].

Fehlberg3 / Heun's method $O(h^3)$

In[95]:=
```
yhF3[n_] := y0 + h/2 (F[y0] + Fy0Taylor[y0 + h F[y0], n - 1])
```

In[96]:=
```
Expand[yhF3[4]]
```

Out[96]=
$$h\,F[y_0]+y_0+\frac{1}{2}h^2\,F[y_0]\,F'[y_0]+\frac{1}{4}h^3\,F[y_0]^2\,F''[y_0]+\frac{1}{12}h^4\,F[y_0]^3\,F^{(3)}[y_0]$$

In[97]:=
```
Series[yhF3[4] - yhF[4], {h, 0, 3}] (* Verified: O(h^3) accurate *)
```

Out[97]=
$$\frac{1}{12}\left(-2\,F[y_0]\,F'[y_0]^2+F[y_0]^2\,F''[y_0]\right)h^3+O[h]^4$$

Fehlberg4 $O(h^4)$

In[98]:=
```
yhF4[n_] := y₀ + h/6 (F[y₀] + Fy0Taylor[y₀ + h F[y₀], n - 1] +
    4 Fy0Taylor[y₀ + h/4 (F[y₀] + Fy0Taylor[y₀ + h F[y₀], n - 2]) , n - 1])
```

In[99]:=
```
Series[yhF4[4] - yhF[4], {h, 0, 4}] (* Verified: O(h^4) accurate *)
```

Out[99]= $-\frac{1}{24}\left(F[y_0]\,F'[y_0]^3\right) h^4 + O[h]^5$

RK4 $O(h^5)$

In[100]:=
```
k1[n_] := Fy0Taylor[y₀, n]
```

In[101]:=
```
k2[n_] := Fy0Taylor[y₀ + h/2 k1[n - 1], n] (* O(h^n) accurate expansion --
   NB: Only O(h^(n-1)) need for inner gradient since multiplied by h *)
```

In[102]:=
```
k3[n_] := Fy0Taylor[y₀ + h/2 k2[n - 1], n]
```

In[103]:=
```
k4[n_] := Fy0Taylor[y₀ + h k3[n - 1], n]
```

In[104]:=
```
k1[3]
```

Out[104]= $F[y_0]$

In[105]:=
```
k2[3]
```

Out[105]= $F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0] + \frac{1}{48} h^3\,F[y_0]^3\,F^{(3)}[y_0]$

In[106]:=
```
k3[3]
```

Out[106]= $F[y_0] + \frac{1}{2} h\,F'[y_0]\left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right) + \frac{1}{8} h^2\,F''[y_0]\left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right)^2 + \frac{1}{48} h^3\left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right)^3 F^{(3)}[y_0]$

In[107]:= `k4[3]`

Out[107]=
$$F[y_0]+h F'[y_0]\left(F[y_0]+\frac{1}{2} h F'[y_0]\left(F[y_0]+\frac{1}{2} h F[y_0] F'[y_0]\right)+\frac{1}{8} h^2\left(F[y_0]+\frac{1}{2} h F[y_0] F'[y_0]\right)^2 F''[y_0]\right)+\frac{1}{2} h^2 F''[y_0]\left(F[y_0]+\frac{1}{2} h F'[y_0]\left(F[y_0]+\frac{1}{2} h F[y_0] F'[y_0]\right)+\frac{1}{8} h^2\left(F[y_0]+\frac{1}{2} h F[y_0] F'[y_0]\right)^2 F''[y_0]\right)^2+\frac{1}{6} h^3\left(F[y_0]+\frac{1}{2} h F'[y_0]\left(F[y_0]+\frac{1}{2} h F[y_0] F'[y_0]\right)+\frac{1}{8} h^2\left(F[y_0]+\frac{1}{2} h F[y_0] F'[y_0]\right)^2 F''[y_0]\right)^3 F^{(3)}[y_0]$$

In[108]:=
```
yhRK4 = y0 + h/6 (k1[4] + 2 k2[4] + 2 k3[4] + k4[4]) (* O(h^5) accurate RK4 expansion *)
```

$$\text{yhRK4} = y_0 + \frac{h}{6}\left(k1[4] + 2\,k2[4] + 2\,k3[4] + k4[4]\right) \quad (*\ O\left(h^5\right) \text{ accurate RK4 expansion } *)$$

Out[108]=

$$\begin{aligned}
&y_0 + \\
&\frac{1}{6} h \Bigg(2\,F[y_0] + h\,F'[y_0] \bigg(F[y_0] + \frac{1}{2} h\,F'[y_0] \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right) + \frac{1}{8} \\
&\qquad h^2\,F''[y_0] \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right)^2 + \\
&\qquad \frac{1}{48} h^3 \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right)^3 F^{(3)}[y_0]\bigg) + \\
&\quad \frac{1}{2} h^2\,F''[y_0] \bigg(F[y_0] + \frac{1}{2} h\,F'[y_0] \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right) + \\
&\qquad \frac{1}{8} h^2\,F''[y_0] \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right)^2 + \\
&\qquad \frac{1}{48} h^3 \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right)^3 F^{(3)}[y_0]\bigg)^2 + \\
&\quad \frac{1}{6} h^3\,F^{(3)}[y_0] \bigg(F[y_0] + \frac{1}{2} h\,F'[y_0] \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right) + \\
&\qquad \frac{1}{8} h^2\,F''[y_0] \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right)^2 + \\
&\qquad \frac{1}{48} h^3 \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right)^3 F^{(3)}[y_0]\bigg)^3 + \\
&\quad \frac{1}{24} h^4 \bigg(F[y_0] + \frac{1}{2} h\,F'[y_0] \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right) + \\
&\qquad \frac{1}{8} h^2\,F''[y_0] \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right)^2 + \\
&\qquad \frac{1}{48} h^3 \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0]\right)^3 F^{(3)}[y_0]\bigg)^4 F^{(4)}[y_0] + \\
&\quad 2 \bigg(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0] + \frac{1}{48} h^3\,F[y_0]^3\,F^{(3)}[y_0] + \\
&\qquad \frac{1}{384} h^4\,F[y_0]^4\,F^{(4)}[y_0]\bigg) + 2 \bigg(F[y_0] + \\
&\qquad \frac{1}{2} h\,F'[y_0] \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0] + \frac{1}{48} h^3\,F[y_0]^3\,F^{(3)}[y_0]\right) + \\
&\qquad \frac{1}{8} h^2\,F''[y_0] \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0] + \frac{1}{48} h^3\,F[y_0]^3\,F^{(3)}[y_0]\right)^2 + \\
&\qquad \frac{1}{48} h^3\,F^{(3)}[y_0] \\
&\qquad\quad \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0] + \frac{1}{48} h^3\,F[y_0]^3\,F^{(3)}[y_0]\right)^3 + \frac{1}{384} h^4 \\
&\qquad\quad \left(F[y_0] + \frac{1}{2} h\,F[y_0]\,F'[y_0] + \frac{1}{8} h^2\,F[y_0]^2\,F''[y_0] + \frac{1}{48} h^3\,F[y_0]^3\,F^{(3)}[y_0]\right)^4 F^{(4)}[y_0]\bigg)\Bigg)
\end{aligned}$$

In[109]:=
```
Simplify[Series[yhRK4 - yhF[5], {h, 0, 5}]] (* Verified: O(h^5) accurate!! *)
```

Out[109]=

$$\begin{aligned}
&\frac{1}{2880} F[y_0] \Big(-24\,F'[y_0]^4 + 36\,F[y_0]\,F'[y_0]^2\,F''[y_0] + \\
&\quad 2\,F[y_0]^2\,F'[y_0]\,F^{(3)}[y_0] + F[y_0]^2 \left(-6\,F''[y_0]^2 + F[y_0]\,F^{(4)}[y_0]\right)\Big)\,h^5 + O[h]^6
\end{aligned}$$

Stability of Trapezoid Method on Model Equation (x=ha)

In[110]:=
```
Rtrap[x_] := Abs[(1 + x/2)/(1 - x/2)]
```

In[111]:=
```
Plot[{Rtrap[x], 1}, {x, -10, 2}]
```

Out[111]=



Stability of Heun's Method on Model Equation (x=ha)

In[112]:=
```
RHeun[x_] := Abs[1 + x + 1/2 x^2]
```

In[113]:=
```
Plot[{RHeun[x], 1}, {x, -3, 3}]
```

Out[113]=